
// MALTLIB SAMPLE MACRO FOR WORDPERFECT 6.0

DISPLAY(Off!) ONERROR(end) ONCANCEL(end) 

// A MALTLIB macro must begin with a GETSTRING command
GETSTRING(KeyName) 
SWITCH(KeyName)

// cInsert Library macros here.  Each library macro begins 
with a line consisting of () the CASEOF instruction, 
() followed by the MALT-key name in quotes, and 
() terminated with a colon.  Examples follow:

CASEOF "C":     
Justification(Center!) 

CASEOF "D": //Define a 4-level counter named "WPCounter"
//to be used with MALTLIB-1 through MALTLIB-4

CounterCreate("WPCounter";
{Numbers!;LowLetter!;LowRoman!;Numbers!})

CASEOF "H": //Create Header A
DLGINPUT(On!) DISPLAY(On!)
HeaderFooterWatermarkDlg Type("HAC")

CASEOF "I":
AttributeAppearanceOn(Italics!)

CASEOF "K": //Change case of character under cursor
DISPLAY(OFF!) ASSIGN(VAR1; CTON(?RightChar)) 
BlockOn(CharMode!) PosCharNext 
 IF(VAR1>96 & VAR1<123) 
   ConvertCaseUppercase
 ELSE IF(VAR1>64 & VAR1<91) 
   ConvertCaseLowercase
ENDIF ENDIF
BlockOff 
PosCharNext 

CASEOF "L";"T": //Left/Right/Top/Bottom Margins
FormatMarginsDlg

CASEOF "M":     //Memorandum macro
PROMPT("Working...")
DISPLAY(Off!)



CANCEL(Off!)
FileNew //open a new window for the memorandum
   //Turn off reveal codes, if on
CALL(ScreenSave) 
Center Type("M E M O R A N D U M") 
PosLineBeg PosWordNext BlockOn(CharMode!) PosLineEnd
UnderlineSpaces(No!) AttributeAppearanceOn(Bold!;Underline!) 
BlockOff
PosLineVeryEnd
RepeatValue(6) Hrt
BlockOn(CharMode!) PosLineUp PosLineUp PosLineUp 
TabSet(Relative!;{?MarginLeft+.9";TabRight!;?
MarginLeft+1.1";TabLeft!})
BlockOff PosLineEnd
Tab AttributeAppearanceToggle(Bold!) Type("TO:") PosLineVeryEnd 
Tab
DISPLAY(On!)
STATUSPROMPT("Please type recipient's name, then press F7") 
PAUSEKEY(ExitDlg) 
DISPLAY(OFF!) PosLineEnd Hrt Hrt
Tab AttributeAppearanceToggle(Bold!) Type("FROM:") PosLineVeryEnd
Tab 
   //Substitute your name on the next line
Type("[Your name here]") 
PosLineEnd Hrt Hrt 
   //Enter date as text
Tab AttributeAppearanceToggle(Bold!) Type("DATE:") PosLineVeryEnd
Tab DateText Hrt Hrt 
Tab AttributeAppearanceToggle(Bold!) Type("SUBJECT:") 
PosLineVeryEnd Tab 
DISPLAY(On!)
STATUSPROMPT("Please type subject, then press F7") 
PAUSEKEY(ExitDlg) 
PROMPT("Creating header...") 
   //Draw line
DISPLAY(OFF!) PosLineDown 
Advance(AdvanceDown!;.5u)
GraphicsLineCreate GraphicsLineType(Horizontal!) 
GraphicsLineEnd(Save!)
   //Place recipient's name in variable for header
PosDocTop SearchString("TO:") SearchNext(Regular!) PosWordNext 
BlockOn(CharMode!) PosLineEnd ASSIGN(Recipient; ?BlockedText) 
   //Create header
PosDocTop HeaderA(Create!; AllPages!)
Type("Memorandum to "+Recipient) Hrt DateCode Hrt Type("Page ") 
PageNumberDisplay Hrt SubstructureExit Suppress(PageNumbering!; 
HeaderA!; HeaderB!; FooterA!; FooterB!) 



 //Move cursor to end
PosDocBottom Hrt
   //Restore reveal codes, if necessary
CALL(ScreenRestore) 
//End of memorandum macro

CASEOF "R":
FlushRight

CASEOF "S": //Set tabs
DISPLAY(On!)
TabDlg

CASEOF "V": //View document
PrintPreview

CASEOF "Z":
DLGINPUT(On!)
DISPLAY(On!)
FormatPageDlg Type("S")

CASEOF "F6":
AttributeAppearanceToggle(Bold!;Underline!)

CASEOF "1": //to be used after counter has been defined
//with MALTLIB-D

CounterDisplay("WPCounter";1)
Type(". ")
CounterIncrement("WPCounter";1)

CASEOF "2":
Type("(")
CounterDisplay("WPCounter";2)
Type(") ")
CounterIncrement("WPCounter";2)

CASEOF "3":
Type("(")
CounterDisplay("WPCounter";3)
Type(") ")
CounterIncrement("WPCounter";3)

CASEOF "4":
Type("(")
CounterDisplay("WPCounter";4)
Type(") ")



CounterIncrement("WPCounter";4)

CASEOF "CR":
DISPLAY(On!)
DLGINPUT(On!)
FormatMarginsDlg
Type("P")

CASEOF "RT";"XRT":
DISPLAY(On!)
DLGINPUT(On!)
AdvanceDlg
Type("R")

CASEOF "LE";"XLE":
DISPLAY(On!)
DLGINPUT(On!)
AdvanceDlg
Type("L")

CASEOF "UP";"XUP":
DISPLAY(On!)
DLGINPUT(On!)
AdvanceDlg
Type("U")

CASEOF "DO";"XDO":
DISPLAY(On!)
DLGINPUT(On!)
AdvanceDlg
Type("D")

//Optional error message
DEFAULT:
SHOWCODE(ReverseOn!)
SHOWPOSITION(26; 10;"╒═══════════════════════════╕") 
SHOWPOSITION(51; 11; "   │")
SHOWPOSITION(26; 11; "│ MALTLIB-"+ KeyName+" is unassigned") 
SHOWPOSITION(26; 12; "╘═══════════════════════════╛") 
SHOWCODE(ReverseOff!)
WAIT(20) 

ENDSWITCH

LABEL(end) 
STATUSPROMPT("") 
DISPLAY(On!)



QUIT

//The following sub-routines are available for use by all macros 
in the library.

LABEL(ScreenSave) //Remember if Reveal Codes was on; then turn
// them off

ASSIGN (rc; ?RevealCodesActive)
RevealCodes(Off!)
RETURN

LABEL(ScreenRestore) //Restore Reveal Codes
RevealCodes (rc)
RETURN


